R. Brigola, TH Nürnberg Georg Simon Ohm,
Fakultät Angewandte Mathematik, Physik und Allgemeinwissenschaften

Juli 2013

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2013

Teil 3 Grundwissen über die diskrete Fouriertransformation
Beispiele zur diskreten Fouriertransformation (DFT, FFT)
und zur diskreten Cosinustransformation (DCT I und DCT II)

Das Notebook ist mit Mathematica 9 unter Windows 7 erstellt. Dieses und weitere Demo - Notebooks findet man unter folgender URL des Autors:
www.stiftung-swk.de/mathematica

Dieses Notebook ist eine Fortführung des Notebooks Fourierreihen-Teil3.nb, das auf der o.g. URL zur Verfügung steht. Hier soll als Anwendung der DCT vom Typ I die Numerische Quadratur nach Clenshaw-Curtis vorgestellt werden.

---

2. Anwendungsbeispiele von DFT und DCT

2.1 Numerische Integration nach Clenshaw-Curtis als DCT I - Anwendung

In [1], Abschnitt 5.7 wird die Clenshaw-Curtis-Quadratur auf den Seiten 78-82 dargelegt. Wir wollen sie hier als Anwendung der DCT I am Beispiel der Funktion

$$f[\text{t\_}] = e^{t} + 3\,\text{Cos}[24\,t] - t^{6} \quad \text{auf } [-1,1]$$

betrachten.

Die folgende Darstellung wird so gewählt, dass Leser hier schon einen ersten Eindruck von der Verwendung von Vektoren und Matrizen mit *Mathematica* erhalten. *Man lese dazu bei Bedarf die Referenz:*

http://reference.wolfram.com/mathematica/tutorial/LinearAlgebraMatrixAndTensorOperations.html

Wir wählen als Beispiel m=12 für dann insgesamt 2m+1 benutzte Abtastwerte von f für das Verfahren. Wie in [1] gezeigt, wird dann durch die Clenshaw-Curtis-Quadratur statt f(cos(ϕ))sin(ϕ) mit einem trigonometrischen Näherungspolynom für f(cos(ϕ)) über [0,π] integriert, welches die maximale Kreisfrequenz N=2m besitzt, und damit eine Näherung SN(f) für das gewünschte Integral von f über [-1,1] berechnet.

Wie in [1] gezeigt, kann man zur Berechnung der Gewichte w_n in der Quadraturformel für die Integralnäherung

$$\text{SN(f)} = \sum_{n=0}^{m} w_n \left(f(x_n) + f(-x_n)\right)$$

mit x_n=Cos[nπ/N], 0≤n≤m, eine DCT I mit m+1 Summanden verwenden.

Zur Durchführung im Beispiel definieren wir wie in [1] die (m+1)x(m+1) DCT I-Matrix und gehen vor wie in [1] beschrieben (dort S. 79-80). Mit dem nachfolgend definierten Vektor b für die Gewichteberechnung ([1], S. 80) brauchen wir die Transponierte der DCT I-Matrix, die hier dctmatrixtransposed getauft wird.

In[1]:=
```
ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*
```

Remove::rmnsm : There are no symbols matching "Global`*". ≫

Information::nomatch : No symbol matching Global`* found. ≫

In[2]:=
```
m = 12;
dctpart1 = Table[1 / m Cos[π (k - 1) (n - 1) / m], {k, m + 1}, {n, 2, m}];
dctpart2 = Map[Prepend[#, 1 / (2 m)] &, dctpart1];
vector = Table[1 / (2 m) Cos[π (k - 1)], {k, m + 1}];
dctmatrixtransposed = Append[Transpose[dctpart2], vector] ;
(* mit dctmatrixtransposed//
 MatrixForm können Sie die gewohnte Matrizenform als Output darstellen. Tun
  Sie das aber nicht hier bei der Definition von dctmatrixtransposed,
sonst klappt die Matrizenmultiplikation weiter unten mit b nicht *)
```

In[7]:=
```
dctmatrixtransposed // MatrixForm
```

Out[7]//MatrixForm=

$$\begin{pmatrix}
\frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} & \frac{1}{24} \\
\frac{1}{12} & \frac{1+\sqrt{3}}{24\sqrt{2}} & \frac{1}{8\sqrt{3}} & \frac{1}{12\sqrt{2}} & \frac{1}{24} & \frac{-1+\sqrt{3}}{24\sqrt{2}} & 0 & -\frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{24} & -\frac{1}{12\sqrt{2}} & -\frac{1}{8\sqrt{3}} & -\frac{1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{12} & \frac{1}{8\sqrt{3}} & \frac{1}{24} & 0 & -\frac{1}{24} & -\frac{1}{8\sqrt{3}} & -\frac{1}{12} & -\frac{1}{8\sqrt{3}} & -\frac{1}{24} & 0 & \frac{1}{24} & \frac{1}{8\sqrt{3}} & \frac{1}{12} \\
\frac{1}{12} & \frac{1}{12\sqrt{2}} & 0 & -\frac{1}{12\sqrt{2}} & -\frac{1}{12} & -\frac{1}{12\sqrt{2}} & 0 & \frac{1}{12\sqrt{2}} & \frac{1}{12} & \frac{1}{12\sqrt{2}} & 0 & -\frac{1}{12\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{12} & \frac{1}{24} & -\frac{1}{24} & -\frac{1}{12} & -\frac{1}{24} & \frac{1}{24} & \frac{1}{12} & \frac{1}{24} & -\frac{1}{24} & -\frac{1}{12} & -\frac{1}{24} & \frac{1}{24} & \frac{1}{12} \\
\frac{1}{12} & \frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{8\sqrt{3}} & -\frac{1}{12\sqrt{2}} & \frac{1}{24} & \frac{1+\sqrt{3}}{24\sqrt{2}} & 0 & -\frac{1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{24} & \frac{1}{12\sqrt{2}} & \frac{1}{8\sqrt{3}} & -\frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{12} & 0 & -\frac{1}{12} & 0 & \frac{1}{12} & 0 & -\frac{1}{12} & 0 & \frac{1}{12} & 0 & -\frac{1}{12} & 0 & \frac{1}{12} \\
\frac{1}{12} & -\frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{8\sqrt{3}} & \frac{1}{12\sqrt{2}} & \frac{1}{24} & -\frac{1+\sqrt{3}}{24\sqrt{2}} & 0 & \frac{1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{24} & -\frac{1}{12\sqrt{2}} & \frac{1}{8\sqrt{3}} & \frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{12} & -\frac{1}{24} & -\frac{1}{24} & \frac{1}{12} & -\frac{1}{24} & -\frac{1}{24} & \frac{1}{12} & -\frac{1}{24} & -\frac{1}{24} & \frac{1}{12} & -\frac{1}{24} & -\frac{1}{24} & \frac{1}{12} \\
\frac{1}{12} & -\frac{1}{12\sqrt{2}} & 0 & \frac{1}{12\sqrt{2}} & -\frac{1}{12} & \frac{1}{12\sqrt{2}} & 0 & -\frac{1}{12\sqrt{2}} & \frac{1}{12} & -\frac{1}{12\sqrt{2}} & 0 & \frac{1}{12\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{12} & -\frac{1}{8\sqrt{3}} & \frac{1}{24} & 0 & -\frac{1}{24} & \frac{1}{8\sqrt{3}} & -\frac{1}{12} & \frac{1}{8\sqrt{3}} & -\frac{1}{24} & 0 & \frac{1}{24} & -\frac{1}{8\sqrt{3}} & \frac{1}{12} \\
\frac{1}{12} & -\frac{1+\sqrt{3}}{24\sqrt{2}} & \frac{1}{8\sqrt{3}} & -\frac{1}{12\sqrt{2}} & \frac{1}{24} & -\frac{-1+\sqrt{3}}{24\sqrt{2}} & 0 & \frac{-1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{24} & \frac{1}{12\sqrt{2}} & -\frac{1}{8\sqrt{3}} & \frac{1+\sqrt{3}}{24\sqrt{2}} & -\frac{1}{12} \\
\frac{1}{24} & -\frac{1}{24} & \frac{1}{24} & -\frac{1}{24} & \frac{1}{24} & -\frac{1}{24} & \frac{1}{24} & -\frac{1}{24} & \frac{1}{24} & -\frac{1}{24} & \frac{1}{24} & -\frac{1}{24} & \frac{1}{24}
\end{pmatrix}$$

Nun der für die Gewichteberechnung benötigte Vektor b, wie oben gesagt, und damit die Berechnung des Vektors w der Gewichte für die Quadratur. Sie sind sämtlich positiv mit Summe 1.

In[8]:=
```
beta = Table[2, {k, m + 1}]; beta[[1]] = 1; beta[[m + 1]] = 1; beta
```
Out[8]= {1, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 1}

In[9]:=
```
b = Table[beta[[k]] / (1 - 4 (k - 1)^2), {k, m + 1}]
```
Out[9]= $\left\{1, -\frac{2}{3}, -\frac{2}{15}, -\frac{2}{35}, -\frac{2}{63}, -\frac{2}{99}, -\frac{2}{143}, -\frac{2}{195}, -\frac{2}{255}, -\frac{2}{323}, -\frac{2}{399}, -\frac{2}{483}, -\frac{1}{575}\right\}$

In[10]:=
```
w = N[dctmatrixtransposed.b]
```
Out[10]= {0.00173913, 0.0166755, 0.0340258, 0.0500188, 0.0654954, 0.0796553, 0.0925836, 0.103831, 0.113378, 0.120922, 0.126452, 0.129768, 0.0654559}

In[11]:=
```
Sum[w[[k]], {k, 1, m + 1}]
```
Out[11]= 1.

Nun noch die benötigten Abtastwerte der Funktion f und dann die numerische Integration :

In[12]:=
```
f[t_] = Exp[t] + 3 Cos[24 t] - t^6;
y = Table[N[f[Cos[(k - 1) π / (2 m)]] + f[-Cos[(k - 1) π / (2 m)]]], {k, 1, m + 1}]
```
Out[13]= {3.63124, 2.55048, -0.84054, -4.22864, -0.183221, 8.05541, 0.451138, -0.453541, 7.28713, -3.68553, 8.05142, -3.98271, 8.}

Ergebnis: Der nach Clenshaw - Curtis nun berechnete Integralwert SN ist

In[14]:=
```
SN = w.y
```
Out[14]= 1.83855

Zum Vergleich: Man kann das Integral ja exakt lösen

In[15]:=
```
integral = Integrate[f[t], {t, -1, 1}]
```
Out[15]= $-\frac{2}{7} - \frac{1}{e} + e + \frac{\mathrm{Sin}[24]}{4}$

In[16]:=
```
N[%, 16]
```
Out[16]= 1.838293511071661

In[17]:=
```
NIntegrate[f[t], {t, -1, 1}, WorkingPrecision → 16]
```
Out[17]= 1.838293511071661

In[18]:=
```
NIntegrate[f[t], {t, -1, 1},
 Method → {"GlobalAdaptive", Method → "ClenshawCurtisRule"},
 WorkingPrecision → 16]
```
Out[18]= 1.838293511071680

In[19]:=
```
NIntegrate[f[t], {t, -1, 1},
 Method → {"GlobalAdaptive", Method → "LobattoKronrodRule"},
 WorkingPrecision → 16]
```
Out[19]= 1.838293511071659

Die Abweichung unseres Ergebnisses SN vom exakten Wert kommt natürlich dadurch zustande, dass wir bei der Clenshaw-Curtis-Quadratur mit der fixen Anzahl von 2m+1 Abtaststellen ein trigonometrisches Näherungspolynom numerisch integrieren, während in den Beispielen mit NIntegrate *Mathematica* iteriert, um eine gewünschte Genauigkeit zu erhalten. Die Unterschiede auch hier sind durch die jeweils verwendete Integrationsmethode bedingt. Im ersten Fall - mit dem besten Ergebnis - haben wir einfach *Mathematica* die Wahl der Methode überlassen.

Die Quadratur ist bei N+1=2m+1 Stützstellen (engl. nodes) für Polynome bis zum Grad N exakt. Zur Berechnung der Gewichte genügt eine DCT I der Länge m+1 (vgl. [1], 5.7, S. 80).

Nun ein kleines Programmierbeispiel mit *Mathematica*:

Da wir vielleicht auch in anderen Fällen einmal Funktionen mit Hilfe der Clenshaw-Curtis-Quadratur über gewisse Intervalle [a,b] integrieren wollen, fassen wir nun das Verfahren in einem Programm zusammen, in *Mathematica* als ein sog. Module. In unserem Zusammenhang werden die Integranden als stetige, stückweise stetig differenzierbare Funktionen vorausgesetzt.
Wir rufen das Module mit **ccq[g, {a,b}, m, opt]** auf, wobei g den Integranden bezeichnet, [a,b] das Integrationsintervall, 2m den Grad, bis zu dem Polynome exakt integriert werden, und opt die Anzahl der gewünschten Stellen des Ergebnisses sein soll. Gebraucht wird dabei nur noch die Substitutionsregel aus der Integralrechnung für die Transformation zu einem Integral über [-1,1].

Nun also das benötigte kleine Programm und zwei Beispiele damit:

```
In[20]:= ccq[g_, {a_, b_}, m_, opt_] :=
 Module[{f, x, dctpart1, dctpart2, dctmatrixtransposed, beta, be, w, y, dez},
  f[x_] = g[(b - a) / 2 x + (a + b) / 2] (b - a) / 2;
  (* hier wird Substitution benutzt *)
  dctpart1 = Table[1 / m Cos[π (k - 1) (n - 1) / m], {k, m + 1}, {n, 2, m}];
  dctpart2 = Map[Prepend[#, 1 / (2 m)] &, dctpart1];
  vector = Table[1 / (2 m) Cos[π (k - 1)], {k, m + 1}];
  dctmatrixtransposed = Append[Transpose[dctpart2], vector];
  beta = Table[2, {k, m + 1}]; beta[[1]] = 1; beta[[m + 1]] = 1;
  be = Table[beta[[k]] / (1 - 4 (k - 1) ^2), {k, m + 1}];
  dez = opt; w = N[dctmatrixtransposed.be, dez];
  y = Table[N[f[Cos[(k - 1) π / (2 m)]] +
      f[-Cos[(k - 1) π / (2 m)]], dez], {k, 1, m + 1}];
  w.y]
```

Beispiel 1. Wir wählen g wie die schon oben über [-1, 1] integrierte Funktion f, diesmal aber mit m=24, d.h. 49 Abtastwerten für die numerische Integration. Damit erreichen wir im Vergleich mit Ausnahme der letzten Dezimalstelle das gleiche Ergebnis wie bei der numerischen Berechnung aus der exakten Lösung mit 16 Stellen durch *Mathematica* weiter oben.

```
In[21]:= g = Function[{t}, Exp[t] + 3 Cos[24 t] - t^6];

In[22]:= ccq[g, {-1, 1}, 24, 16]

Out[22]= 1.838293511071663
```

Beispiel 2. Wir integrieren das Polynom $p[x] = x^3 - 3\,x^{12} + 10\,x^{15}$ über [1,3]
mit der Clenshaw-Curtis-Quadratur und 20 Ergebnisstellen.
Mit m⩾8 sollte das Ergebnis (bis auf evtl. numerische, durch die Maschinengenauigkeit bedingte Rundungsfehler) exakt sein.

In[23]:=
```
p[x_] = x^3 - 3 x^12 + 10 x^15;
ccq[p, {1, 3}, 8, 20]
```

Out[24]= $2.6536299538461538462 \times 10^7$

Wir testen das Ergebnis, indem wir das Polynom exakt integrieren und das Ergebnis von *Mathematica* als dezimale Näherung mit ebenfalls 20 Stellen ausgeben lassen.

In[25]:=
```
Integrate[p[x], {x, 1, 3}]
```

Out[25]= $\frac{344\,971\,894}{13}$

In[26]:=
```
N[%, 20]
```

Out[26]= $2.6536299538461538462 \times 10^7$

Ich hoffe, die Beispiele haben Ihnen bzgl. der enthaltenen Mathematik und auch bzgl. der gezeigten Möglichkeiten von *Mathematica* ein wenig Spaß gemacht.
Weitere Anwendungsbeispiele von DFT und DCT mit Hilfe dieses Computeralgebra-Systems als Fortsetzung dieses Abschnitts 2 werde ich in nachfolgenden Notebooks vorstellen.

Weiterführende Literatur zum Thema finden Sie in [1] angegeben. Hier seien nur wenige Quellen genannt, auch für Algorithmen zur numerischen Integration von Funktionen auf hoch-dimensionalen Quadern, für welche die Clenshaw-Curtis-Quadratur Verwendung findet. Mit einer Suchmaschinen-Anfrage zu “Clenshaw Curtis quadrature” finden Sie sehr schnell zahlreiche hervorragende weitere Quellen, besonders auch zu effizienteren Implementierungen des Verfahrens als dies hier zu Demo-Zwecken im Zusammenhang mit der DCT gezeigt ist.

[2] *Mathematica*-Dokumentation zu den Methoden und Optionen bei numerischer Integration mit NIntegrate

http://reference.wolfram.com/legacy/v7/tutorial/NIntegrateIntegrationRules.html

[3] M. Hanke-Bourgeois
Grundlagen der Numerischen Mathematik und des Wissenschaftlichen Rechnens, Teubner, Wiesbaden, 2008

[4] E. Novak, K. Ritter, R. Schmitt, A. Steinbauer (1999) On a recent interpolatory method for high-dimensional integration, J. Comput. Appl. Math. 112, 215-228